Order this document
as AN433/D

# AN433

# TV on-screen display using the MC68HC05T1

**By Peter Topping**
**Motorola Ltd., East Kilbride, Scotland**



## INTRODUCTION

The "T" members of the MC68HC05 family of MCUs provide a convenient and cost effective method of adding on-screen-display (OSD) to TVs and VCRs. As well as the OSD capability, they include 8 Kbytes of ROM (adequate for Teletext, frequency-synthesis, stereo and OSD), 320 bytes of RAM, a 16-bit timer and 8 pulse-width-modulated D/A converters. The MC68HC(7)05T7/8 also includes IIC hardware and, by using a 56/64 pin package, 4 ports of I/O independent of the OSD, serial and D/A outputs. It is thus suitable for large full-feature chassis. The MC68HC05T1 is in the middle of the price/performance range and includes most of the features of the MC68HC05T8 but in a 40-pin package. This is achieved by sharing I/O with the other pin functions (SPI, OSD, D/A). Even if all these features are used there is sufficient I/O for most applications. The low cost MC68HC05T4 has 5 Kbytes of ROM and 96 bytes of RAM making it suitable for simpler (mono, non-Teletext) applications.

## 68HC05T1 OSD FEATURES

- Programmable display of 10 rows of 18 characters
- 24 byte (18 data + 6 control) single row architecture
- Settable in software to any one of four standards
- Zero inter-row and inter-column spacing
- 64 user-defined mask-programmable 8 x 13 characters
- Programmable horizontal position
- Character colour selectable from 4 colours/row
- Software programmable (start, stop and colour) window
- 4 character sizes (normal, double height and/or width)
- Half-dot character rounding
- Selectable half-dot black outline.

## OSD CHARACTERISTICS

The HC05TX series have an OSD capability of 10 rows of 18 characters. Each row can contain characters of four colours selected from the eight available colours (black, blue, green, cyan, red, magenta, yellow and white). The rows can independently select double height and/or double width and the start and stop positions of a background window of any colour. The signals sent to the TV are Red, Green, Blue, fast-blanking and half-tone. Separate horizontal and vertical synchronisation inputs are required.

The OSD architecture employed includes only a single line of display RAM. This makes the software more complicated but reduces the silicon area required to implement the OSD function. The software is required to update the display RAM on a regular basis. When operating in the 625-line PAL standard the updates must occur at 1.66 ms (26 lines) intervals in order to display adjacent lines. The OSD hardware can generate an interrupt when an update is required. There are 18 data registers (one for each character) and 6 control registers arranged as shown below. The table is for the T1; some of the control bits are different in the T4/7/8.

| | | |
|---|---|---|
| $20 - $31 | OSD | Data registers. |
| $32 | CAS | Read: status, Write: colours 1 & 2 and outline enable. |
| $33 | C34 | Colours 3 & 4. |
| $34 | RAD | Row address, character size, int. enable, RGB invert. |
| $35 | WCR | OSD & PLL enable, Window enable and start column. |
| $36 | CCP | Window colour and end column. |
| $37 | HPD | Horizontal position, standard selection. |



freescale™ semiconductor

The OSD display is timed from an on-chip 14 MHz oscillator which is phase locked to the TV's line synchronisation pulses. The vertical synchronisation depends on the standard in use. Four standards are available (15.75 kHz/60 Hz, 31.5 kHz/120 Hz, 15.625 kHz/50 Hz and 31.25 kHz/100 Hz). The standard is selected by control bits in the T1/2 but is automatic in the T4 and the T7/8.

64 OSD characters are mask programmed along with the user ROM. The spacing and full size of the characters is the same at 8 x 13 (for 625-line standard). This allows continuous graphics. Half-dot interpolation hardware doubles the apparent resolution to produce smoother characters. A software selectable black outline (a half-dot wide) is also implemented in the hardware. Because the half-dot circuitry has to know the information for the next line of pixels, a 14th line is available in the character generator ROM to facilitate look ahead. The vertical height of a character is 26 lines (52 including interlace) and the horizontal width is 2 2/7 µs (1/7 µs per half pixel).

## SOFTWARE

There are several approaches to writing OSD software to operate with the single line architecture. The choice will affect the amount of ROM and RAM used. One principle is to have a separate interrupt routine for each type of row to be displayed. This method will use little RAM but will be inefficient in its use of ROM. The other approach is to write a single interrupt routine which transfers display information from a block of normal RAM to the display RAM as it is required for each new line. This method will be more ROM efficient but requires a RAM location for every display character. The amount of RAM used depends on the maximum amount of data which has to be displayed at any one time. The choice between these two methods will depend on the type of data to be displayed. The first method may be better if much of the displayed data is fixed. This could be, for example, a series of menus. The second method will however be more appropriate if the data is mostly variable. This will usually be the case in conventional TV applications.

This application note describes an implementation of the second of the above approaches. A block of RAM is used to contain a copy of all the data to be displayed. The size of this block can be changed to reflect the number of rows and the number of characters per row. The choice made in the example described here is 8 rows of 16 characters. This is slightly less than the maximum available and was chosen because the total number of characters (128) corresponds to the available page 1 RAM in the MC68HC05T1. The choice of 16 characters per row also slightly simplifies the software. The software allows any eight of the ten available rows to be used but only the first 16 of the 18 available characters. This choice does not prevent access to the right-hand-side of the screen as the display can be moved to the right under software control. The use of page 1 for the RAM does not incur any significant compromise in execution time. It also leaves free the page 0 RAM for the rest of the TV control software, which would be made less efficient if it had to use page 1 RAM, where direct addressing and bit manipulation instructions cannot be used. This choice slightly increases the ROM used by the OSD code, as 3-byte extended store instructions sometimes need to be used to write data to the RAM used for OSD characters.

The 1-byte indexed addressing mode can however be used in page 1. This addressing mode can access up to address $1FE and is made use of in the example software. For example the OSDCLR routine used to initialise RAM locations used for OSD employs a CLR DRAM-1,X instruction. DRAM is the start of page 1 RAM at $100 so DRAM-1 evaluates as $FF a 1-byte offset.

## INTERRUPT ROUTINE

The OSD update interrupt routine (NLINE) shown in the program listing transfers data from page 1 RAM to display RAM each time an interrupt occurs. The first operation is to increment the pointer which selects the next row number. This pointer (OSDL) is subsequently used to transfer the appropriate data from page 1 RAM to the OSD RAM. So that any row number can be used the pointer selects the number from a table unique to each type of display. The appropriate table is determined by the value of LIND. The pointer is incremented until the corresponding row number is zero when the pointer is reset to zero. This allows any sub-set of up to 8 of the 10 available rows to be used. The next row number (ORed with the character size information contained in RAM) is written into the appropriate register ($34). The row number in this register is compared by the OSD hardware with the current position of the raster. When they match, an interrupt is generated and the next interrupt routine is performed. The other control registers are then updated from the page 0 RAM locations, which are used for this purpose.

To save RAM only three (RAD, CAS _& CCR) of the six control registers are loaded in this way. The pointer OSDL is multiplied by 3 using the table M3, as this is quicker than shifting and adding. In this example the other registers are loaded by the main program and therefore have fixed values for each display. The fixed registers are Colour 3/4 ($33), Window enable/start column ($35) and Horizontal position delay ($37). As this choice would not allow windows to be enabled on individual rows, window enable is controlled by the un-used bit (6) in the RAM byte used to update the Colour 1/2 register (CAS). This choice of fixed registers limits the flexibility of the display but clearly all registers can be updated on a line-by-line basis if more RAM is used. The limitations imposed by this choice are that colours 3 & 4, the window start column and the horizontal position apply to a whole display rather than to individual lines. In practice these constraints were not found to be significant restrictions for the displays required for TV use.

The interrupt routine then transfers the relevant OSD data from page 1 RAM into the OSD data registers. This is done using linear, repetitive code in order to minimise the time taken by the interrupt routine. The code used uses 8 cycles (4 µs) for each byte transferred. Less ROM space would be utilised if a loop was employed but this would use 28 cycles per byte. The best choice depends on whether time or ROM use is more critical. The example code includes a cycle count to calculate the length of the interrupt routine. The time taken is 121 ± 4 µs. This includes the time taken by the interrupt itself. An alternative method of OSD data transfer (TOSD2) using a loop is included as comments in the listing. It would take an additional 165 µs.

The last task performed by the interrupt routine is to control any character or window flashing. The software allows one or two characters (on a selected row) and one window (on the same or a different row) to be flashed at a rate determined by the MCU's timer. This function could be performed outwith the interrupt routine in the main program and the time taken to perform it is not included in the figure given above.

## MAIN OSD PROGRAM

The remainder of the OSD control program does not write directly to most of the display registers. It simply puts the required display and control information into the blocks of RAM allocated for this purpose, together with supplying the co-ordinates of any required flashing characters or windows. It must, however, write to the display control registers not updated by the interrupt routine; in this example these are $33, $35 and $37. The program has 4 main parts. These are the idle, channel name table, program/channel number and analogue displays. The idle display applies when no transient display (eg program number and channel number or name) is on. The OSD idle condition is selectable between blank and a small program number at the bottom right hand corner of the screen.

The OSD example program (assembler listing included) is just part of the code required to control a TV set. This program was incorporated in HC05T1 software along with four other modules. These were the base module (idle loop, transient control, local keyboard, IR, IIC and reset), the tuning module (PLL, analogue and NVM control), the stereo module (stereoton and Nicam) and the Teletext module (FLOF level 1.5).

The microprocessor in a TV application will usually need to handle the reception of IR commands. Polled methods of IR reception are most effective if the time made unavailable to them by interrupts is minimised. It is for this reason that the illustrated OSD interrupt routine was written to execute as fast as possible. This is, however, not so much of a problem if the TCAP facility is used for IR reception. When a falling edge occurs, the timer value is saved and it does not matter if the interrupt which processes this information is not serviced until several hundred microseconds later. The allowable size of this delay will of course depend on the IR protocol in use. The bi-phase protocol used with the example OSD software (transmitter chip: MC144105) has a minimum spacing of 1 ms between consecutive edges.

The next section describes the OSD features of this software. Some of the data used in the OSD is passed from other modules (particularly the tuning module). The same RAM allocation file was used in all modules so this part of the listing shows the locations used to pass data between them.

## OSD FEATURES PROVIDED IN THE EXAMPLE PROGRAM

### Program change

When keys 0-9, PC- or PC+ are pressed, the new program number appears (in cyan) at the bottom-right-hand corner of the screen in double height/double width characters and stays for 5 seconds after the last change. Above this display either the channel name (if one has been defined) or the channel number is shown (normal size). After 5 seconds this display times out and there is either no display or a permanent normal size program number display. This is selectable using the Teletext MIX key.

For program numbers of 10 and over, three keys are required. They are selected by first pressing "—". Two flashing dashes are displayed, the first 0-9 key (only 0-4 valid) will be taken as the tens digit and the second as the units digit. If a new program number has not been selected within 30 seconds, the TV returns to the previous display (nothing or the old program number).

### Channel mode

When the P/C key is pressed, the program number and channel name (or number) is displayed for 5 seconds as an indication of the current status. If it is pressed again during this period, the TV changes to channel mode. This will remain for 30 seconds after the last key-press. The display (in yellow) shows the program number as in program mode along with the channel number. The channel number flashes to show that it will be changed if a number or PC- or PC+ key is pressed. New channels can be selected. If the STORE key is pressed then the current channel is stored against the current program number. If no key is pressed for 30 seconds, the TV returns to program mode. If the channel has been changed but STORE not pressed then the TV will retune back to the channel stored against the current program number.

### Automatic search

When SEARCH is pressed the TV goes into the channel mode and the on screen display is as described above. The channel number is incremented at a rate of 2 per second until a signal is found. The search then stops. A press of STORE returns the TV to program mode, storing the new channel against the current program number.

### Analogues

When any of the analogues are selected the appropriate logo is displayed along with a horizontal bar indicating the current value in the D/A convertor (full-scale 63). Display returns to default (nothing or program number) 5 seconds after the last change. If no analogue is selected the volume is shown (and adjusted) when the ANALOGUE +/- keys are used.

### Channel name table

Up to 24 channels can have a 4 character name and standard bit associated with them. If the channel number and standard of one of these entries in the table correspond to those selected by the current program number then the name is displayed along with the program number when the program is selected or when P/C is pressed. Entry of names is done using the Teletext INDEX key. When it is pressed a table of six lines is displayed. Each line (identified by a "station" number in the leftmost column) contains a channel number, standard and the associated name. All of this data is user definable.

One character on the screen flashes to indicate the current position of the cursor. The character at the cursor position can be changed through 0-9, A-Z and space by pressing PC+ or PC- (0-9 for channel number digits and PAL/SECAM for standard). When a character (or the standard) is changed, its colour changes from yellow to red. The cursor is moved to the left and right by the Teletext RED and GREEN keys and up and down by the BLUE and YELLOW keys. The current line appears in a light blue (cyan) window as opposed to the dark blue window used for the other lines. The whole table scrolls when the cursor is required to go beyond the bottom (or top) of the current display.

To save a name the STORE key is pressed. This will save the name and standard on the current line against its channel number. This is indicated by the colour of any changed characters returning to yellow. Any changes which have been made to lines other than the one being stored are lost. Channel 00 cannot have a name. The procedure for removing a name from the table is to set the channel number to zero and then to save the line. Any name left on the line will not be used. The table display is exited by pressing the Teletext INDEX key. The function of each key is shown at the bottom of the display.

## EXAMPLE PROGRAM

```
                        ****************************************************************************
                        *                                                                          *
                        *             TV/Teletext/OSD/Stereo program (MC68HC05T1/8).               *
                        *                                                                          *
                        *                       On-Screen-Display module                           *
                        *                                                                          *
                        * This software was developed by Motorola Ltd. for demonstration purposes. *
                        *  No liability can be accepted for its use in any specific application.   *
                        *       Original software copyright Motorola - all rights reserved.        *
                        *                                                                          *
                        *              P. Topping                      25th May '90                *
                        *                                                                          *
                        ****************************************************************************



                                IMPORT  CHEX,CBCD,READ,WRITE,CDISP2

                                EXPORT  NLINE,PCOSD,DRAM,ANOSD,PRDSP,CHST
                                EXPORT  CUP,CDWN,CLFT,CRGT,PLUS,MINUS,SAVE,OSDLE,OSDEF

                                LIB     RAMT1.S05

                                SECTION.S .RAM,COMM
                        ******************************************************
                        *                                                    *
                        *       Teletext RAM allocation.                     *
                        *                                                    *
                        ******************************************************

23 00000000             SUB1    RMB     1
23 00000001             R1      RMB     1       mode register
23 00000002             R2      RMB     1       page request address register
23 00000003             R3      RMB     1       page req. data reg. col. 0 : mag.
23 00000004             C1      RMB     1       "     "    "    "    "   1 : pgt.
23 00000005             C2      RMB     1       "     "    "    "    "   2 : pgu.
23 00000006             C3      RMB     1       "     "    "    "    "   3 : ht.
23 00000007             C4      RMB     1       "     "    "    "    "   4 : hu.
23 00000008             C5      RMB     1       "     "    "    "    "   5 : mt.
23 00000009             C6      RMB     1       "     "    "    "    "   6 : mu.
23 0000000a             SUB2    RMB     1
23 0000000b             R4      RMB     1       display chapter register
23 0000000c             R5      RMB     1       display control register (normal)
23 0000000d             R6      RMB     1       display control register (news/sub)
23 0000000e             R7      RMB     1       display mode register
23 0000000f             SUB3    RMB     1
23 00000010             R8      RMB     1       active chapter register
23 00000011             R9      RMB     1       active row register
23 00000012             R10     RMB     1       active column register
23 00000013             R11     RMB     1       active data register
23 00000014             PH      RMB     1       2nd     "       "
23 00000015             PT      RMB     1       3rd     "       "
23 00000016             PU      RMB     1       4th     "       "
23 00000017             LIFO    RMB     9       LINKED PAGE No. LIFO BUFFER
23 00000020             PAGE    RMB     7       PAGE No. INPUT BUFFER
23 00000027             PAG0    RMB     3       AC0 PAGE No.
23 0000002a             PAG1    RMB     3       AC1 PAGE No.
23 0000002d             PAG2    RMB     3       AC2 PAGE No.
23 00000030             PAG3    RMB     3       AC3 PAGE No.
23 00000033             PAGC    RMB     3       CYAN PAGE No.
23 00000036             PAGI    RMB     3       INDEX PAGE No.
23 00000039             PDP     RMB     1       PAGE DIGIT POINTER
23 0000003a             ACC     RMB     4       DISP, RED, GREEN, YELLOW AC. CIR.
23 0000003e             WACC    RMB     1       WORKING ACC No.
23 0000003f             ADDR    RMB     1       IIC ADDRESS
23 00000040             DPNT    RMB     1       IIC DATA POINTER FOR WRITE
23 00000041             SUBADR  RMB     1       IIC SUB-ADDRESS
23 00000042             IOBUF   RMB     4       IIC BUFFER, +2 & +3 RSRVD FOR PLL
23 00000046             STAT2   RMB     1       0: ROW24 FETCH FLAG
23                      *                       1: REMOTE REPEATING
23                      *                       2: SEARCH/STANDBY IIC LOCK
23                      *                       3: STANDBY STATUS
23                      *                       4: UPDATE PENDING
23                      *                       5: DIFFERENCE FOUND
23                      *                       6: NO TELETEXT TRANSMISSION
23                      *                       7: MIXED
23 00000047             LINKC   RMB     1       LINK OPTIONS
23 00000048             STAT3   RMB     1       0: CYAN   LINK ON
23                      *                       1: YELLOW LINK ON
23                      *                       2: GREEN  LINK ON
23                      *                       3: LINKS/ROW24  ON
```

```
23                              ********************************************************
23                              *                                                      *
23                              *       General RAM allocation.                        *
23                              *                                                      *
23                              ********************************************************
23
23 00000049                     PLLHI   RMB     1       PLL DIVIDE RATIO MSB
23 0000004a                     PLLOW   RMB     1       PLL DIVIDE RATIO LSB
23 0000004b                     W1      RMB     1       WORKING
23 0000004c                     W2      RMB     1       "
23 0000004d                     W3      RMB     1       "
23 0000004e                     COUNT   RMB     1       LOOP COUNTER
23 0000004f                     KOUNT   RMB     1       LOCAL KEYBOARD COUNTER
23 00000050                     CNT     RMB     1       12.8mS (inc, free running)
23 00000051                     CNT1    RMB     1       12.8mS (inc, reset every 1S during transient)
23                              *CNT2   RMB     1       3.25 S (inc, store timeout)
23 00000052                     CNT3    RMB     1       3.25 S (dec, automatic standby timeout)
23 00000053                     CNT4    RMB     1       12.8mS (cleared for row24 delay when page arrives)
23 00000054                     CNT5    RMB     1       12.8mS (inc, transient mute)
23 00000055                     TMR     RMB     1       TRANSIENT DISPLAY SECONDS COUNTER
23 00000056                     STAT    RMB     1       0: TV/TELETEXT
23                              *                       1: IIC R/W
23                              *                       2: HOLD
23                              *                       3: IR REPEAT INHIBIT
23                              *                       4: TRANSIENT DISPLAY ON
23                              *                       5: TIME HOLD
23                              *                       6: SUB-PAGE MODE
23                              *                       7: IR TASK PENDING
23 00000057                     STAT4   RMB     1       0: KEY FUNCTION PERFORMED
23                              *                       1: LOCAL REPEATING
23                              *                       2: P/C PROG : 0, CHAN : 1
23                              *                       3: MUTE (TRANSIENT)
23                              *                       4: OSD STATUS TRANSIENT
23                              *                       5: MUTE (BUTTON)
23                              *                       6: COINCIDENCE MUTE
23                              *                       7: SEARCH
23 00000058                     PWR     RMB     1       $55 AT RESET, $AA NORMALLY
23 00000059                     PROG    RMB     1       CURRENT PROGRAM NUMBER
23 0000005a                     CHAN    RMB     1       CURRENT CHANNEL NUMBER
23 0000005b                     DISP    RMB     1       CURRENT DISPLAY NUMBER
23 0000005c                     FTUNE   RMB     1       FINE TUNING REGISTER
23 0000005d                     AVOL    RMB     1       VOLUME LEVEL
23 0000005e                     KEY     RMB     1       CODE OF PRESSED KEY (LOCAL)
23 0000005f                     NUM0    RMB     4       LED DISPLAY RAM
23 00000063                     IRRA1   RMB     1       IR INTERRUPT TEMP.
23 00000064                     IRRA2   RMB     1       "       "       "
23 00000065                     IRRA3   RMB     1       "       "       "
23 00000066                     IRRA4   RMB     1       "       "       "
23 00000067                     DIFFH   RMB     1       IR TIME DIFFERENCE
23 00000068                     DIFFL   RMB     1       "    "      "
23 00000069                     IRH     RMB     1       IR CODE BIT
23 0000006a                     IRL     RMB     1       COLLECTION
23 0000006b                     IRCODE  RMB     1
23 0000006c                     IRCNT   RMB     1
23 0000006d                     IRCMCT  RMB     1
23 0000006e                     OLDIR   RMB     1
23                              ********************************************************
23                              *                                                      *
23                              *       RAM allocation for Stereoton.                  *
23                              *                                                      *
23                              ********************************************************
23
23 0000006f                     POLLTM  RMB     1       Poll timer
23 00000070                     TONEA   RMB     1       Tone (unadjusted for loudness)
23 00000071                     LBAL    RMB     1       Loudspeaker balance variable
23 00000072                     LVL     RMB     1       Loudspeaker left volume                 (reg 1)
23 00000073                     LVR     RMB     1       Loudspeaker right volume                (reg 2)
23 00000074                     HVL     RMB     1       Headphone volume left                   (reg 3)
23 00000075                     HVR     RMB     1       Headphone volume right                  (reg 4)
23 00000076                     TONE    RMB     1       Tone variable (Bass/Treble)             (reg 5)
23 00000077                     MATRIX  RMB     1       Current matrix                          (reg 6)
23 00000078                     MATNO   RMB     1       Present mode (mono/stereo/lan 1/l1.l2/l2.l1)
23 00000079                     WS1     RMB     1       Workspace 1  (no interrupt useage
23 0000007a                     WS2     RMB     1       Workspace 2   for these please..)
23
23 0000007b                     VAV     RMB     1
23
23 0000007c                     MONCNT  RMB     1       Mono ident count          Ident detection
23 0000007d                     STECNT  RMB     1       Stereo ident count        variables
23 0000007e                     DULCNT  RMB     1       Dual lang ident count
23 0000007f                     ERRCNT  RMB     1       Error ident count
23 00000080                     RCOUNT  RMB     1       Ident countdown
23 00000081                     RANGE   RMB     1       Total ident poll number
23
23 00000082                     TEMP    RMB     1
```

```
23
23 00000083                    STAT5   RMB     1       0: LOUDNESS
23                             *                       1: VCR
23                             *                       2: OSD NAME TABLE
23                             *                       3: OSD DEFAULT P/C NUMBER
23                             *                       4: ANALOGUE OSD ON
23                             *                       5: NAME-TABLE STANDARD
23                             *                       6: STANDARD CHANGED
23                             *                       7: RE-INITIALISE TELETEXT
23
23 00000084                    STAT6   RMB     1       0: 2-DIGIT PROGRAM ENTRY
23
23 00000085                    TMPRG   RMB     1       TEMPORARY PROGRAM NUMBER
23                             ********************************************************
23                             *                                                      *
23                             *       OSD RAM allocation.                            *
23                             *                                                      *
23                             ********************************************************
23
23 00000086                    CAS1    RMB     1       ROW 1, colour 1/2 & outline enable
23 00000087                    RAD1    RMB     1       Row address & character size
23 00000088                    CCR1    RMB     1       Window colour & end column
23 00000089                    CAS2    RMB     1       ROW 2, colour 1/2 & outline enable
23 0000008a                    RAD2    RMB     1       Row address & character size
23 0000008b                    CCR2    RMB     1       Window colour & end column
23 0000008c                    CAS3    RMB     1       ROW 3, colour 1/2 & outline enable
23 0000008d                    RAD3    RMB     1       Row address & character size
23 0000008e                    CCR3    RMB     1       Window colour & end column
23 0000008f                    CAS4    RMB     1       ROW 4, colour 1/2 & outline enable
23 00000090                    RAD4    RMB     1       Row address & character size
23 00000091                    CCR4    RMB     1       Window colour & end column
23 00000092                    CAS5    RMB     1       ROW 5, colour 1/2 & outline enable
23 00000093                    RAD5    RMB     1       Row address & character size
23 00000094                    CCR5    RMB     1       Window colour & end column
23 00000095                    CAS6    RMB     1       ROW 6, colour 1/2 & outline enable
23 00000096                    RAD6    RMB     1       Row address & character size
23 00000097                    CCR6    RMB     1       Window colour & end column
23 00000098                    CAS7    RMB     1       ROW 7, colour 1/2 & outline enable
23 00000099                    RAD7    RMB     1       Row address & character size
23 0000009a                    CCR7    RMB     1       Window colour & end column
23 0000009b                    CAS8    RMB     1       ROW 8, colour 1/2 & outline enable
23 0000009c                    RAD8    RMB     1       Row address & character size
23 0000009d                    CCR8    RMB     1       Window colour & end column
23
23 0000009e                    OSDL    RMB     1       CURRENT OSD ROW POINTER
23 0000009f                    LIND    RMB     1       ROW TABLE INDEX
23 000000a0                    BROW    RMB     1       CHARACTER FLASH ROW
23 000000a1                    BCOL    RMB     1       CHATACTER FLASH COLUMNS
23 000000a2                    WROW    RMB     1       WINDOW FLASH ROW
23                             *ROW1   RMB     1       FIRST ROW No. (NAME TABLE)
23
23 000000a3                    ANAL    RMB     1
23 000000a4                    ANAF    RMB     1
23
23 000000a5                    TEMP2   RMB     1
23
23 000000a6                            RMB     3       UNUSED
23
23 000000a9                    STACK   RMB     22      23 BYTES USED FOR STACK
23 000000bf                    SP      RMB     1       (1 INTERRUPT AND 9 NESTED SUBS)
23
23
23 00000000                    KEYI    EQU     $00
23 00000000                    KEYO    EQU     $00
23 00000000                    KEYIO   EQU     $00
23 00000003                    SERO    EQU     $03
23
23 0000000a                    VOLU    EQU     $0A     D/A 2   JP08 IN EVB
23 0000000b                    CONT    EQU     $0B     D/A 3   JP09 IN EVB
23 0000000c                    BRIL    EQU     $0C     D/A 4   JP10 IN EVB
23 0000000d                    SATU    EQU     $0D     D/A 5   JP11 IN EVB
23
23 00000005                    L1      EQU     $05     Lang. 1 indicator bit       (LEDOUT)
23 00000006                    L2      EQU     $06     Lang. 2 indicator bit       (LEDOUT)
23 00000004                    WIDE    EQU     $04     Wide matrix bit             (MATRIX+LEDOUT)
23 00000005                    PST     EQU     $05     Pseudo-stereo matrix bit    (  "      "   )
23 00000006                    VCR     EQU     $06     VCR active bit              (STAT3+LEDOUT)
23 00000005                    LOUD    EQU     $05     LOUDness effect active bit  (STAT3)
23 00000003                    MUT     EQU     $03     Mute indicator bit          (MATRIX+LEDOUT)
23
23 00000080                    STADR   EQU     $80     Stereoton address  (IIC)
23 00000019                    NORMVOL EQU     &25     normal volume (mid balance)
23
```

```
23                                      ********************************************************
23                                      *                                                      *
23                                      *       Equates.                                       *
23                                      *                                                      *
23                                      ********************************************************
23
23 00000000                             PORTA   EQU     $00     Port A address
23 00000001                             PORTB   EQU     $01     Port B    "
23 00000002                             PORTC   EQU     $02     Port C    "
23 00000003                             PORTD   EQU     $03     Port D    "
23 00000004                             DDRA    EQU     $04     Port A data direction reg.
23 00000005                             DDRB    EQU     $05     Port B   "      "      "
23 00000006                             DDRC    EQU     $06     Port C   "      "      "
23 00000007                             DDRD    EQU     $07     Port D   "      "      "
23
23 00000012                             TCR     EQU     $12     Timer control register.
23 00000013                             TSR     EQU     $13     Timer status register.
23 00000014                             ICRH    EQU     $14     Input capture register, high.
23 00000015                             ICRL    EQU     $15     Input capture register, low.
23 00000016                             OCRH    EQU     $16     Output compare register, high.
23 00000017                             OCRL    EQU     $17     Output compare register, low.
23 00000018                             TDRH    EQU     $18     Timer data register, high.
23 00000019                             TDRL    EQU     $19     Timer data register, low.
23 0000001c                             MISC    EQU     $1C     Misc. register
23
24
25 00000020                             OSD     EQU     $20     18 OSD data registers
26 00000032                             CAS     EQU     $32     Color & status register
27 00000033                             C34     EQU     $33     Color 3/4 register
28 00000034                             RAD     EQU     $34     Row address & character size
29 00000035                             WCR     EQU     $35     Window/Column register
30 00000036                             CCR     EQU     $36     Column/color register
31 00000037                             HPD     EQU     $37     Horizontal position delay
32
33 00000039                             MAD     EQU     $39     M-bus address register
34 0000003a                             MFD     EQU     $3A     M-bus frequency divider
35 0000003b                             MCR     EQU     $3B     M-bus control register
36 0000003c                             MSR     EQU     $3C     M-bus status register
37 0000003d                             MDR     EQU     $3D     M-bus data register
38 0000003e                             TR1     EQU     $3E     Test 1, OSD/Timer/PLM
39 0000003f                             TR2     EQU     $3F     Test 2, EPROM
40
41                                              SECTION .RAM2
42
43 00000000                             DRAM    RMB     128
44
45                                              SECTION .ROM2
47                                      ********************************************************
48                                      *                                                      *
49                                      *       OSD update routine - row number & data.        *
50                                      *                                                      *
51                                      ********************************************************
52
53 00000000 >b600                       NLINE   LDA     OSDL            3               INCREMENT
54 00000002  4c                                 INCA                    3   6           LINE POINTER
55 00000003 >b700                       STAG    STA     OSDL            4  10       27
56 00000005 >bb00                               ADD     LIND            3  13       30
57 00000007  97                                 TAX                     2  15       32
58 00000008 >d60000                             LDA     LTAB0,X         5  20       37
59 0000000b  27f6                               BEQ     STAG            3  23  OR  40     32 +/- 8
60 0000000d >be00                               LDX     OSDL            3               LINE POINTER
61 0000000f >de0000                             LDX     M3,X            5   8           MULTIPLY BY 3
62 00000012 >ea00                               ORA     RAD1,X          4  12           CHARACTER SIZE INFO.
63 00000014  b734                               STA     RAD             4  16
64 00000016 >e600                               LDA     CAS1,X          4  20
65 00000018  b732                               STA     CAS             4  24
66 0000001a  1f35                               BCLR    7,WCR           5  29
67 0000001c  49                                 ROLA                    3  32           GET BIT 6
68 0000001d  49                                 ROLA                    3  35           OF CASx
69 0000001e  2402                               BCC     SKIPW           3  38
70 00000020  1e35                               BSET    7,WCR           5  43           USE IT TO ENABLE WINDOW
71 00000022 >e600                       SKIPW   LDA     CCR1,X          4  47
72 00000024  b736                               STA     CCR             4  51           83 +/- 8
73 00000026 >be00                       LLOK    LDX     OSDL            3   7           LINE POINTER
74 00000028 >de0000                             LDX     M16,X           5  12           MULTIPLY BY 16
75
76 0000002b >e6f0                       TOSD1   LDA    <DRAM-16,X       4               GET DATA AND WRITE
77 0000002d  b720                               STA     OSD             4   8           IT TO OSD REGISTER
78 0000002f >e6f1                               LDA    <DRAM-15,X
79 00000031  b721                               STA     OSD+1
80 00000033 >e6f2                               LDA    <DRAM-14,X                       FAST OSD DATA TRANSFER
81 00000035  b722                               STA     OSD+2                           TAKES ONLY 128 (8x16)
82 00000037 >e6f3                               LDA    <DRAM-13,X                       CYCLES.
```

```
 83 00000039  b723                      STA     OSD+3
 84 0000003b >e6f4                      LDA    <DRAM-12,X
 85 0000003d  b724                      STA     OSD+4
 86 0000003f >e6f5                      LDA    <DRAM-11,X
 87 00000041  b725                      STA     OSD+5
 88 00000043 >e6f6                      LDA    <DRAM-10,X
 89 00000045  b726                      STA     OSD+6
 90 00000047 >e6f7                      LDA    <DRAM-9,X
 91 00000049  b727                      STA     OSD+7
 92 0000004b >e6f8                      LDA    <DRAM-8,X
 93 0000004d  b728                      STA     OSD+8
 94 0000004f >e6f9                      LDA    <DRAM-7,X
 95 00000051  b729                      STA     OSD+9
 96 00000053 >e6fa                      LDA    <DRAM-6,X
 97 00000055  b72a                      STA     OSD+10
 98 00000057 >e6fb                      LDA    <DRAM-5,X
 99 00000059  b72b                      STA     OSD+11
100 0000005b >e6fc                      LDA    <DRAM-4,X
101 0000005d  b72c                      STA     OSD+12
102 0000005f >e6fd                      LDA    <DRAM-3,X
103 00000061  b72d                      STA     OSD+13
104 00000063 >e6fe                      LDA    <DRAM-2,X
105 00000065  b72e                      STA     OSD+14
106 00000067 >e6ff                      LDA    <DRAM-1,X                140          223 +/- 8
107 00000069  b72f                      STA     OSD+15                   WITH INT 242 (121 +/- 4 US)
108
109                             *TOSD2  STX     TMP1            4              ÀLTERNATIVE OSD DATA
110                             *       LDX     #16             2   6          TRANSFER USING A LOOP.
111                             *       STX     TMP2            4  10          THIS HAS THE ADVANTAGE
112                             *OSDOOP LDX     TMP1            3              OF USING 44 BYTES LESS
113                             *       LDA    <DRAM-1,X        4   7          ROM BUT IT USES TWO MORE
114                             *       DEC     TMP1            5  12          TEMPORARY RAM LOCATIONS
115                             *       LDX     TMP2            3  15          AND TAKES 330 CYCLES
116                             *       STA    <OSD-1,X         5  20          (165uS) LONGER.
117                             *       DEC     TMP2            5  25
118                             *       BNE     OSDOOP          3  28          28x16+10=458=128+330
119
120                             ********************************************************
121                             *                                                      *
122                             *       Character and window flash.                    *
123                             *                                                      *
124                             ********************************************************
125
126 0000006b >09001f                    BRCLR   4,CNT,WBLK      5  25
127 0000006e >b600              CHBLK   LDA     BROW            3  28          CHARACTER BLINK
128 00000070  2719                      BEQ     NCHBK           3  31
129 00000072  b634                      LDA     RAD             3  34
130 00000074  a40f                      AND     #$0F            2  36
131 00000076 >b100                      CMP     BROW            3  39
132 00000078  2611                      BNE     NCHBK           3  42
133 0000007a >b600                      LDA     BCOL            3  45
134 0000007c  a40f                      AND     #$0F            3  48
135 0000007e  97                        TAX                     2  50
136 0000007f  ad1b                      BSR     SPFL           34  84          1st CHARACTER (LS NIBBLE)
137 00000081 >be00                      LDX     BCOL            3  87
138 00000083  54                        LSRX                    3  90
139 00000084  54                        LSRX                    3  93
140 00000085  54                        LSRX                    3  96
141 00000086  54                        LSRX                    3  99
142 00000087  2702                      BEQ     NCHBK           3 102          IF MS NIBBLE ZERO THEN NO
143 00000089  ad11                      BSR     SPFL           34 136          2nd CHARACTER
144 0000008b  200e              NCHBK   BRA     NOBLK           3 139
145 0000008d >b600              WBLK    LDA     WROW                           WINDOW BLINK
146 0000008f  270a                      BEQ     NOBLK
147 00000091  b634                      LDA     RAD
148 00000093  a40f                      AND     #$0F
149 00000095 >b100                      CMP     WROW
150 00000097  2602                      BNE     NOBLK                          371 +/- 8
151 00000099  1f35                      BCLR    7,$35                          with INT 381 +/- 8 cycles
152 0000009b  80                NOBLK   RTI                     9 148          ie 190.5 +/- 4 us
153
154 0000009c  e620              SPFL    LDA     OSD,X       6 + 4  10
155 0000009e  a43f                      AND     #$3F            2  12
156 000000a0  2609                      BNE     NTSP            3  15
157 000000a2  e620                      LDA     OSD,X           4  19
158 000000a4  a4c0                      AND     #$C0            2  21
159 000000a6  ab0e                      ADD     #$0E            2  23
160 000000a8  e720                      STA     OSD,X           5  28
161 000000aa  81                        RTS                     6  34
162 000000ab  6f20              NTSP    CLR     OSD,X
163 000000ad  81                        RTS
```

```
                                ********************************************************
                                *                                                      *
                                *       OSD idle condition.                            *
                                *                                                      *
                                ********************************************************

000000ae >060004                OSDEF   BRSET   3,STAT5,DOFF
000000b1 >1600                          BSET    3,STAT5
000000b3  2002                          BRA     OSDLE

000000b5 >1700                  DOFF    BCLR    3,STAT5

000000b7  9b                    OSDLE   SEI
000000b8 >1900                          BCLR    4,STAT5         NOT ANALOGS
000000ba >1500                          BCLR    2,STAT5         NOT NAME TABLE

000000bc  ae1d                          LDX     #29             CLEAR PAGE 0
000000be >6fff                  DOOP    CLR     CAS1-1,X        OSD CONTROL
000000c0  5a                            DECX                    BYTES
000000c1  26fb                          BNE     DOOP

000000c3 >cd0000                        JSR     CDISP2
000000c6  3f30                          CLR     $30
000000c8  3f31                          CLR     $31
000000ca >06000a                        BRSET   3,STAT5,SKPDEF
000000cd >c6000c                        LDA     DRAM+12         PROGRAM NUMBER
000000d0  b730                          STA     $30
000000d2 >c6000e                        LDA     DRAM+14
000000d5  b731                          STA     $31
000000d7  5f                    SKPDEF  CLRX
000000d8  a67f                          LDA     #127
000000da  ad1b                          BSR     OSDCLR

000000dc  a620                          LDA     #%00100000      HORIZONTAL POSITION : ZERO
000000de  b737                          STA     HPD
000000e0  a6a3                          LDA     #%10100011      COLOR 1,0 = RED, CYAN, EDGE ON
000000e2 >050002                        BRCLR   2,STAT4,PMD     PROGRAM MODE ?
000000e5  a6a6                          LDA     #%10100110      NO, COLOR 0 = YELLOW
000000e7 >b700                  PMD     STA     CAS1
000000e9  a610                          LDA     #%00010000      SINGLE WIDTH/HIGHT
000000eb >b700                          STA     RAD1

000000ed  a60c                          LDA     #$0C            DEFAULT TO VOLUME
000000ef >b700                          STA     ANAL
000000f1 >a600                          LDA     #AVOL
000000f3 >b700                          STA     ANAF
000000f5  9a                            CLI
000000f6  81                            RTS

000000f7  4c                    OSDCLR  INCA
000000f8 >b700                          STA     W1
000000fa  5c                    DCLR    INCX
000000fb >6fff                          CLR     DRAM-1,X
000000fd >b300                          CPX     W1
000000ff  26f9                          BNE     DCLR
00000101  81                            RTS
```

```
222                                  ******************************************************
223                                  *                                                    *
224                                  *       Program/Channel/Name display.                *
225                                  *                                                    *
226                                  ******************************************************
227
228 00000102  3334002328003334   BNTAB   FCB     $33,$34,0,$23,$28,0,$33,$34            ST CH ST
229 0000010a  240e00002e212d25           FCB     $24,$0E,0,0,$2E,$21,$2D,$25,$C0        D.  NAME
230 00000113  005c009e003c00fe   MTAB    FCB     0,$5C,0,$9E,0,$3C,0,$FE                > < Y ^
231 0000011b  008b008d006d00a9           FCB     0,$8B,0,$8D,0,$6D,0,$A9,$C0            + - M I
232
233 00000124 >040090             PRDSP   BRSET   2,STAT5,OSDLE
234 00000127 >1400                       BSET    2,STAT5
235 00000129  a61a                       LDA     #%00011010      COLOR 2 = GREEN
236 0000012b  b733                       STA     C34             COLOR 3 = CYAN
237 0000012d  a6e1                       LDA     #%11100001      OSD & PLL ON,
238 0000012f  b735                       STA     WCR             WINDOW ON (COLUMN 1)
239 00000131  a621                       LDA     #%00100001      HORIZONTAL POSITION : ONE
240 00000133  b737                       STA     HPD
241 00000135  3f30                       CLR     $30             PUT A SPACE AT 17th AND 18th
242 00000137  3f31                       CLR     $31             CHARACTERS
243
244 00000139  a6e5                       LDA     #%11100101      COLOR 1,0 = RED, MAGENTA, EDGE ON
245 0000013b >b700                       STA     CAS1
246 0000013d  a6e6                       LDA     #%11100110      AND WINDOW ON (USING BIT 6)
247 0000013f >b700                       STA     CAS8
248
249 00000141  a610                       LDA     #%00010000      SINGLE WIDTH/HIGHT, INTERRUPTS ON
250 00000143  ae18                       LDX     #24
251 00000145 >e7fd               STLP    STA     RAD1-3,X
252 00000147  5a                         DECX
253 00000148  5a                         DECX
254 00000149  5a                         DECX
255 0000014a  26f9                       BNE     STLP
256
257 0000014c  a6e6                       LDA     #%11100110      COLOR 1,0 = RED, YELLOW, EDGE ON
258 0000014e  ae12                       LDX     #18
259 00000150 >e7fd               STLP2   STA     CAS2-3,X
260 00000152  5a                         DECX
261 00000153  5a                         DECX
262 00000154  5a                         DECX
263 00000155  26f9                       BNE     STLP2
270
271 00000157 >3f00                       CLR     OSDL
272 00000159  a609                       LDA     #LTAB3-LTAB0
273 0000015b >b700                       STA     LIND            THIRD TABLE
274
275 0000015d  a603                       LDA     #3              START AT ROW 3
276 0000015f >b700                       STA     BROW
277 00000161  a603                       LDA     #$03            START AT COLUMN 3
278 00000163 >b700                       STA     BCOL
279 00000165  a600                       LDA     #0
280 00000167 >b700                       STA     WROW
281 00000169  a601                       LDA     #1
282 0000016b >b700                       STA     COUNT
283
284 0000016d  ae80                       LDX     #128            CLEAR 1st THRU 8th ROWS
285 0000016f >6fff               ACLR    CLR     DRAM-1,X
286 00000171  5a                         DECX
287 00000172  26fb                       BNE     ACLR
288
289 00000174  5f                         CLRX
290 00000175 >d60000             BNPL    LDA     BNTAB,X
291 00000178  a1c0                       CMP     #$C0
292 0000017a  2706                       BEQ     FINBN
293 0000017c >d70000                     STA     DRAM,X
294 0000017f  5c                         INCX
295 00000180  20f3                       BRA     BNPL
296
297 00000182  ae10               FINBN   LDX     #16
298 00000184 >1d00                       BCLR    6,STAT5         CLEAR STANDARD CHANGE FLAG
299 00000186 >b600                       LDA     COUNT
300 00000188 >b700                       STA     W2
```

```
302                                  ********************************************************
303                                  *                                                      *
304                                  *       Program/Channel/Name main loop.                *
305                                  *                                                      *
306                                  ********************************************************
307
308 0000018a >b600                   BNLP    LDA     W2              STATION No.
309 0000018c >bf00                           STX     W3
310 0000018e >cd0000                         JSR     CBCD
311 00000191 >b700                           STA     W1
312 00000193  a40f                           AND     #$0F
313 00000195  ab10                           ADD     #$10
314 00000197 >be00                           LDX     W3
315 00000199 >d70001                         STA     DRAM+1,X
316 0000019c >b600                           LDA     W1
317 0000019e  44                             LSRA
318 0000019f  44                             LSRA
319 000001a0  44                             LSRA
320 000001a1  44                             LSRA
321 000001a2  2602                           BNE     NOTZR
322 000001a4  a6f0                           LDA     #$F0            LEADING ZERO BLANK
323 000001a6  ab10                   NOTZR   ADD     #$10
324 000001a8 >d70000                 STLSN   STA     DRAM,X
325 000001ab >b600                           LDA     W2              STATION No.
326 000001ad  abdf                           ADD     #$DF
327 000001af >b700                           STA     SUBADR
328 000001b1  a6a0                           LDA     #$A0
329 000001b3 >b700                           STA     ADDR
330 000001b5 >cd0000                         JSR     READ
331 000001b8 >b601                           LDA     IOBUF+1         CHANNEL No.
332 000001ba  a47f                           AND     #$7F
333 000001bc >cd0000                         JSR     CBCD
334 000001bf >b700                           STA     W1
335 000001c1  a40f                           AND     #$0F
336 000001c3  ab10                           ADD     #$10
337 000001c5 >be00                           LDX     W3
338 000001c7 >d70004                         STA     DRAM+4,X        MSD
339 000001ca >b600                           LDA     W1
340 000001cc  44                             LSRA
341 000001cd  44                             LSRA
342 000001ce  44                             LSRA
343 000001cf  44                             LSRA
344 000001d0  ab10                           ADD     #$10
345 000001d2 >d70003                         STA     DRAM+3,X        LSD
347                                  ********************************************************
348                                  *                                                      *
349                                  *       Standard and bottom line.                      *
350                                  *                                                      *
351                                  ********************************************************
352
353 000001d5 >1a00                           BSET    5,STAT5
354 000001d7 >0e0102                         BRSET   7,IOBUF+1,PALS
355 000001da >1b00                           BCLR    5,STAT5
356
357 000001dc >cd0000                 PALS    JSR     CHGST
358
359 000001df >b600                           LDA     W2
360 000001e1 >cd0000                         JSR     GNAME2
361 000001e4  9f                             TXA
362 000001e5  ab10                           ADD     #16
363 000001e7  97                             TAX
364 000001e8 >3c00                           INC     W2
365 000001ea  a360                           CPX     #96
366 000001ec  2203                           BHI     NOJMP
367 000001ee >cc0000                         JMP     BNLP
368
369 000001f1  5f                     NOJMP   CLRX
370 000001f2 >d60000                 MTL     LDA     MTAB,X
371 000001f5  a1c0                           CMP     #$C0
372 000001f7  2706                           BEQ     ANFIN
373 000001f9 >d70070                         STA     DRAM+112,X
374 000001fc  5c                             INCX
375 000001fd  20f3                           BRA     MTL
376
377 000001ff >cd0000                 ANFIN   JSR     WIND
378 00000202 >1800                   SEC30   BSET    4,STAT
379 00000204  a61e                           LDA     #30
380 00000206 >b700                           STA     TMR
381 00000208  81                             RTS
```

```
383                                 ******************************************************
384                                 *                                                    *
385                                 *       Look for channel name.                       *
386                                 *                                                    *
387                                 ******************************************************
388
389 00000209 >04002f  FNAME   BRSET   2,STAT4,NONAME          CHANNEL MODE
390 0000020c  a6a0          LDA    #$A0
391 0000020e >b700          STA    ADDR
392 00000210  a6e0          LDA    #$E0
393 00000212 >b700          STA    SUBADR
394
395 00000214 >b600          LDA    CHAN
396 00000216 >cd0000        JSR    CHEX
397 00000219 >b700          STA    COUNT
398 0000021b  030205        BRCLR  1,PORTC,OLOOP           38.9 MHz ?
399 0000021e  0b0202        BRCLR  5,PORTC,OLOOP           NO, SECAM ?
400 00000221 >1e00          BSET   7,COUNT                 NO, PAL
401
402 00000223 >cd0000  OLOOP   JSR    READ
403 00000226 >b601          LDA    IOBUF+1
404 00000228  020202        BRSET  1,PORTC,IF38            38.9 MHz ?
405 0000022b  a47f          AND    #$7F                    YES, SO IGNORE STANDARD
406 0000022d  2704    IF38    BEQ    CH0
407 0000022f >b100          CMP    COUNT                   CHAN
408 00000231  274d          BEQ    NOFND
409 00000233 >3c00    CH0     INC    SUBADR
410 00000235 >b600          LDA    SUBADR
411 00000237  a1f7          CMP    #$F7
412 00000239  23e8          BLS    OLOOP
413 0000023b >be00    NONAME  LDX    W3
414 0000023d  a623          LDA    #$23                    NO NAME SO DISPLAY Ch. No.
415 0000023f >d70008        STA    DRAM+8,X
416 00000242  a628          LDA    #$28
417 00000244 >d70009        STA    DRAM+9,X
418 00000247 >b600          LDA    CHAN
419 00000249  44            LSRA
420 0000024a  44            LSRA
421 0000024b  44            LSRA
422 0000024c  44            LSRA
423 0000024d  ab10          ADD    #$10
424 0000024f >d7000a        STA    DRAM+10,X               3rd CHAR (NAME)
425 00000252 >b600          LDA    CHAN
426 00000254  a40f          AND    #$0F
427 00000256  ab10          ADD    #$10
428 00000258 >d7000b        STA    DRAM+11,X               4th CHAR (NAME)
429
430 0000025b  a630          LDA    #$30            P
431 0000025d >d7000d        STA    DRAM+13,X
432 00000260  a621          LDA    #$21            A
433 00000262 >d7000e        STA    DRAM+14,X
434 00000265  a62c          LDA    #$2C            L
435 00000267 >d7000f        STA    DRAM+15,X
436 0000026a  030212        BRCLR  1,PORTC,SPAL            38.9 MHz ?
437 0000026d  0a020f        BRSET  5,PORTC,SPAL            NO, PAL ?
438 00000270  a633          LDA    #$33            S       NO, SECAM
439 00000272 >d7000d        STA    DRAM+13,X
440 00000275  a625          LDA    #$25            E
441 00000277 >d7000e        STA    DRAM+14,X
442 0000027a  a623          LDA    #$23            C
443 0000027c >d7000f        STA    DRAM+15,X
444 0000027f  81      SPAL    RTS
445
446 00000280 >b600    NOFND   LDA    SUBADR
447 00000282  a0df            SUB    #$DF
448 00000284  48      GNAME2  LSLA                   x2
449 00000285  48              LSLA                   x4
450 00000286  ab7c            ADD    #$7C
451 00000288 >b700            STA    SUBADR
452 0000028a >cd0000          JSR    READ
453 0000028d >be00            LDX    W3
454 0000028f >b601            LDA    IOBUF+1
455 00000291 >d7000c          STA    DRAM+12,X               1st CHAR (NAME)
456 00000294 >b600            LDA    IOBUF
457 00000296 >d7000d          STA    DRAM+13,X               2nd CHAR (NAME)
458 00000299 >3c00            INC    SUBADR
459 0000029b >3c00            INC    SUBADR
460 0000029d >cd0000          JSR    READ
461 000002a0 >be00            LDX    W3
462 000002a2 >b601            LDA    IOBUF+1
463 000002a4 >d7000e          STA    DRAM+14,X               3rd CHAR (NAME)
464 000002a7 >b600            LDA    IOBUF
465 000002a9 >d7000f          STA    DRAM+15,X               4th CHAR (NAME)
466 000002ac  81              RTS
467
```

```
468                                     ********************************************************
469                                     *                                                      *
470                                     *       Cursor control (left & right).                 *
471                                     *                                                      *
472                                     ********************************************************
473
474 000002ad  03040c0d0e0f           CURTAB  FCB     3,4,12,13,14,15
475
476 000002b3  ad19                  CLFT    BSR     FCUR
477 000002b5  a305                          CPX     #5
478 000002b7  2502                          BLO     NRAP1
479 000002b9  aeff                          LDX     #$FF
480 000002bb  5c                    NRAP1   INCX
481 000002bc >d60000                NEWC    LDA     CURTAB,X
482 000002bf >b700                          STA     BCOL
483 000002c1 >cc0000                SEC32   JMP     SEC30
484
485 000002c4  ad08                  CRGT    BSR     FCUR
486 000002c6  5d                            TSTX
487 000002c7  2602                          BNE     NRAP2
488 000002c9  ae06                          LDX     #6
489 000002cb  5a                    NRAP2   DECX
490 000002cc  20ee                          BRA     NEWC
491
492 000002ce >b600                  FCUR    LDA     BCOL
493 000002d0 >b700                          STA     W1
494 000002d2  aeff                          LDX     #$FF
495 000002d4  5c                    CRNF    INCX
496 000002d5 >d60000                        LDA     CURTAB,X
497 000002d8 >b100                          CMP     W1
498 000002da  26f8                          BNE     CRNF
499 000002dc  81                            RTS
500
501 000002dd  a631                  WIND    LDA     #%00110001      WINDOW BLUE, OFF AT 17
502
503 000002df  ae12                          LDX     #18
504 000002e1 >e7fd                  STLP3   STA     CCR2-3,X
505 000002e3  5a                            DECX
506 000002e4  5a                            DECX
507 000002e5  5a                            DECX
508 000002e6  26f9                          BNE     STLP3
509
510 000002e8  a611                          LDA     #%00010001      WINDOW BLACK, OFF AT 17
511 000002ea >b700                          STA     CCR1
512 000002ec >b700                          STA     CCR8
513 000002ee >be00                          LDX     BROW
514 000002f0  5a                            DECX
515 000002f1  5a                            DECX
516 000002f2 >de0000                        LDX     M3,X
517 000002f5 >e600                          LDA     CCR1,X
518 000002f7 >b700                          STA     W2
519 000002f9 >1c00                          BSET    6,W2
520 000002fb >b600                          LDA     W2
521 000002fd >e700                          STA     CCR1,X
522 000002ff  81                            RTS
523
524                                     ********************************************************
525                                     *                                                      *
526                                     *       Cursor control (up & down).                    *
527                                     *                                                      *
528                                     ********************************************************
529
530 00000300 >b600                  CUP     LDA     BROW
531 00000302  a103                          CMP     #3
532 00000304  2304                          BLS     TOOSM
533 00000306 >3a00                          DEC     BROW
534 00000308  20d3                          BRA     WIND
535 0000030a >b600                  TOOSM   LDA     COUNT
536 0000030c  a101                          CMP     #1
537 0000030e  27b1                  SEC31   BEQ     SEC32
538 00000310 >3a00                          DEC     COUNT
539 00000312  2012                          BRA     FIN30
540
541 00000314 >b600                  CDWN    LDA     BROW
542 00000316  a108                          CMP     #8
543 00000318  2404                          BHS     TOOBG
544 0000031a >3c00                          INC     BROW
545 0000031c  20bf                          BRA     WIND
546 0000031e >b600                  TOOBG   LDA     COUNT
547 00000320  a113                          CMP     #19
548 00000322  27ea                          BEQ     SEC31
549 00000324 >3c00                          INC     COUNT
550 00000326 >cd0000                FIN30   JSR     SEC30
551 00000329 >cc0000                        JMP     FINBN
```

```
                                ********************************************************
                                *                                                      *
                                *       Standard change.                               *
                                *                                                      *
                                ********************************************************

0000032c >1b00                  CHST    BCLR    5,STAT5         DEFAULT TO SECAM
0000032e >be00                          LDX     BROW
00000330  5a                            DECX
00000331  5a                            DECX
00000332  5a                            DECX
00000333 >de0000                        LDX     M16,X
00000336 >d60006                        LDA     DRAM+6,X
00000339  a43f                          AND     #$3F
0000033b  a130                          CMP     #$30            PAL ?
0000033d  2702                          BEQ     SZER
0000033f >1a00                          BSET    5,STAT5         NO, MAKE IT PAL
00000341 >1c00                  SZER    BSET    6,STAT5         STANDARD CHANGED
00000343 >cd0000                        JSR     SEC30

00000346  a630                  CHGST   LDA     #$30
00000348 >d70006                        STA     DRAM+6,X
0000034b  a621                          LDA     #$21
0000034d >d70007                        STA     DRAM+7,X
00000350  a62c                          LDA     #$2C
00000352 >d70008                        STA     DRAM+8,X
00000355  a600                          LDA     #0
00000357 >d70009                        STA     DRAM+9,X
0000035a >d7000a                        STA     DRAM+10,X
0000035d  03021c                        BRCLR   1,PORTC,PAL             38.9MHz ?
00000360 >0a0019                        BRSET   5,STAT5,PAL             NO, PAL ?
00000363  a633                  SECAM   LDA     #$33                    NO, SECAM
00000365 >d70006                        STA     DRAM+6,X
00000368  a625                          LDA     #$25
0000036a >d70007                        STA     DRAM+7,X
0000036d  a623                          LDA     #$23
0000036f >d70008                        STA     DRAM+8,X
00000372  a621                          LDA     #$21
00000374 >d70009                        STA     DRAM+9,X
00000377  a62d                          LDA     #$2D
00000379 >d7000a                        STA     DRAM+10,X
0000037c >0d0012                PAL     BRCLR   6,STAT5,NSTCH

0000037f  9f                            TXA
00000380  ab05                          ADD     #5
00000382 >b700                          STA     COUNT
00000384 >d60006                XLP     LDA     DRAM+6,X
00000387  ab40                          ADD     #$40
00000389 >d70006                        STA     DRAM+6,X
0000038c  5c                            INCX
0000038d >b300                          CPX     COUNT
0000038f  26f3                          BNE     XLP

00000391  81                    NSTCH   RTS
                                ********************************************************
                                *                                                      *
                                *       Character change.                              *
                                *                                                      *
                                ********************************************************

00000392  ad40                  PLUS    BSR     GETIT
00000394  4c                            INCA
00000395  a43f                          AND     #$3F

00000397  a119                          CMP     #$19            9
00000399  2208                          BHI     MT9
0000039b  a110                  LTE9    CMP     #$10            0
0000039d  2410                          BHS     NLT0
0000039f  a610                          LDA     #$10            0
000003a1  200c                          BRA     NLT0
000003a3  a121                  MT9     CMP     #$21            A
000003a5  2202                          BHI     MTA
000003a7  a621                          LDA     #$21            A
000003a9  a13a                  MTA     CMP     #$3A            Z
000003ab  2302                          BLS     NLT0
000003ad  a600                  SPACE   LDA     #$00            SPACE

000003af  aa40                  NLT0    ORA     #$40
000003b1 >d70000                        STA     DRAM,X
000003b4 >cc0000                        JMP     SEC30

000003b7  ad1b                  MINUS   BSR     GETIT
000003b9  4a                            DECA
000003ba  a43f                          AND     #$3F
```

```
639 000003bc  a121                      CMP     #$21            A
640 000003be  2508                      BLO     LTA
641 000003c0  a13a              GTEA    CMP     #$3A            Z
642 000003c2  23eb                      BLS     NLT0
643 000003c4  a63a                      LDA     #$3A            Z
644 000003c6  20e7                      BRA     NLT0
645 000003c8  a119              LTA     CMP     #$19            9
646 000003ca  2302                      BLS     LT9
647 000003cc  a619                      LDA     #$19            9
648 000003ce  a110              LT9     CMP     #$10            0
649 000003d0  24dd                      BHS     NLT0
650 000003d2  20d9                      BRA     SPACE
651
652 000003d4 >b600              GETIT   LDA     BROW
653 000003d6  a002                      SUB     #2
654 000003d8  48                        LSLA                    x2
655 000003d9  48                        LSLA                    x4
656 000003da  48                        LSLA                    x8
657 000003db  48                        LSLA                    x16
658 000003dc >bb00                      ADD     BCOL
659 000003de  97                        TAX
660 000003df >d60000                    LDA     DRAM,X
661 000003e2  81                        RTS
663                             ********************************************************
664                             *                                                      *
665                             *       Name store.                                    *
666                             *                                                      *
667                             ********************************************************
668
669 000003e3  a6a0              SAVE    LDA     #$A0
670 000003e5 >b700                      STA     ADDR
671 000003e7 >b600                      LDA     COUNT
672 000003e9 >bb00                      ADD     BROW
673 000003eb  48                        LSLA
674 000003ec  48                        LSLA
675 000003ed  ab70                      ADD     #$70
676 000003ef >b700                      STA     SUBADR
677 000003f1  a603                      LDA     #3
678 000003f3 >b700                      STA     W1
679 000003f5 >b700                      STA     W2
680 000003f7 >be00                      LDX     BROW
681 000003f9  5a                        DECX
682 000003fa  5a                        DECX
683 000003fb  58                        LSLX
684 000003fc  58                        LSLX
685 000003fd  58                        LSLX
686 000003fe  58                        LSLX
687 000003ff >bf00                      STX     W3
688 00000401 >d6000c                    LDA     DRAM+12,X
689 00000404  a43f                      AND     #$3F
690 00000406 >b700                      STA     IOBUF
691 00000408 >d6000d                    LDA     DRAM+13,X
692 0000040b  a43f                      AND     #$3F
693 0000040d >b701                      STA     IOBUF+1
694 0000040f >ae00                      LDX     #SUBADR
695 00000411 >cd0000                    JSR     WRITE
696
697 00000414 >3c00                      INC     SUBADR
698 00000416 >3c00                      INC     SUBADR
699 00000418 >be00                      LDX     W3
700 0000041a  a603                      LDA     #3
701 0000041c >b700                      STA     W1
702 0000041e >b700                      STA     W2
703 00000420 >d6000e                    LDA     DRAM+14,X
704 00000423  a43f                      AND     #$3F
705 00000425 >b700                      STA     IOBUF
706 00000427 >d6000f                    LDA     DRAM+15,X
707 0000042a  a43f                      AND     #$3F
708 0000042c >b701                      STA     IOBUF+1
709 0000042e >ae00                      LDX     #SUBADR
710 00000430 >cd0000                    JSR     WRITE
```

```
                                ********************************************************
                                *                                                      *
                                *       Name store (continued).                        *
                                *                                                      *
                                ********************************************************

00000433 >be00                          LDX     W3
00000435 >d60003                        LDA     DRAM+3,X
00000438  48                            LSLA
00000439  48                            LSLA
0000043a  48                            LSLA
0000043b  48                            LSLA
0000043c >b700                          STA     W1
0000043e >d60004                        LDA     DRAM+4,X
00000441  a40f                          AND     #$0F
00000443 >bb00                          ADD     W1
00000445 >cd0000                        JSR     CHEX
00000448 >b700                          STA     IOBUF

0000044a >be00                          LDX     W3
0000044c >d60006                        LDA     DRAM+6,X
0000044f  a43f                          AND     #$3F
00000451  a133                          CMP     #$33
00000453  2702                          BEQ     STSEC
00000455 >1e00                          BSET    7,IOBUF

00000457 >b600                  STSEC   LDA     COUNT
00000459 >bb00                          ADD     BROW
0000045b  abdc                          ADD     #$DC
0000045d >b700                          STA     SUBADR
0000045f  a602                          LDA     #2
00000461 >b700                          STA     W1
00000463 >b700                          STA     W2
00000465 >ae00                          LDX     #SUBADR
00000467 >cd0000                        JSR     WRITE

0000046a >cd0000                        JSR     SEC30
0000046d >cc0000                        JMP     FINBN

                                ********************************************************
                                *                                                      *
                                *       OSD line number tables.                        *
                                *                                                      *
                                ********************************************************

00000470  0a00                  LTAB0   FCB     10,0                    IDLE DISPLAY
00000472  090800                LTAB1   FCB     9,8,0                   PR/CH DISPLAY
00000475  07080a00              LTAB2   FCB     7,8,10,0                ANALOGUE DISPLAY
00000479  0203040506070809      LTAB3   FCB     2,3,4,5,6,7,8,9,0       PR/CH/STD/NAME TABLE

00000482  1020304050607080      M16     FCB     $10,$20,$30,$40,$50,$60,$70,$80         MULT x 16
0000048a  000306090c0f1215      M3      FCB     0,3,6,9,12,15,18,21                     MULT x  3
```

```
765                                 ********************************************************
766                                 *                                                      *
767                                 *       Bottom corner Program/Channel no. display.     *
768                                 *                                                      *
769                                 ********************************************************
770
771 00000492  a60c                 PCOSD   LDA     #$0C
772 00000494 >b700                         STA     ANAL
773 00000496 >a600                         LDA     #AVOL
774 00000498 >b700                         STA     ANAF
775 0000049a >1900                         BCLR    4,STAT5         NOT ANALOGS
776 0000049c  a60a                         LDA     #%00001010      COLOR 2 = GREEN
777 0000049e  b733                         STA     C34             COLOR 3 = BLUE
778 000004a0  a670                         LDA     #%01110000      OSD & PLL ON,
779 000004a2  b735                         STA     WCR             WINDOW OFF (COLUMN 16)
780 000004a4  a622                         LDA     #%00100010      HORIZONTAL POSITION : TWO
781 000004a6  b737                         STA     HPD
782 000004a8  3f30                         CLR     $30             PUT A SPACE AT 17th AND 18th
783 000004aa  3f31                         CLR     $31             CHARACTERS
784 000004ac  5f                           CLRX
785 000004ad  a609                         LDA     #9
786 000004af >cd0000                       JSR     OSDCLR          CLEAR UNUSED CHARACTERS
787 000004b2  ae10                         LDX     #16
788 000004b4  a61f                         LDA     #31
789 000004b6 >cd0000                       JSR     OSDCLR          CLEAR UNUSED CHARACTERS
790 000004b9  a610                 PNAME   LDA     #16
791 000004bb >b700                         STA     W3
792 000004bd >b600                         LDA     PROG
793 000004bf  2703                         BEQ     SKPGN
794 000004c1 >cd0000                       JSR     FNAME
795 000004c4  a601                 SKPGN   LDA     #1              START AT 1 TO PREVENT
796 000004c6 >b700                         STA     OSDL            DOUBLE-HIGHT-LINE-SHIFT FLASH
797 000004c8  a602                         LDA     #LTAB1-LTAB0
798 000004ca >b700                         STA     LIND            FIRST TABLE
799 000004cc  a6a3                         LDA     #%10100011      COLOR 1,0 = RED, CYAN, EDGE ON
800 000004ce >050002                       BRCLR   2,STAT4,PMD2    PROGRAM MODE ?
801 000004d1  a6a6                         LDA     #%10100110      NO, COLOR 0 = YELLOW
802 000004d3 >b700                 PMD2    STA     CAS1
803 000004d5 >b700                         STA     CAS2
804 000004d7  a6d0                         LDA     #%11010000      DOUBLE WIDTH/HIGHT
805 000004d9 >b700                         STA     RAD1
806 000004db  a610                         LDA     #%00010000      SINGLE WIDTH/HIGHT
807 000004dd >b700                         STA     RAD2
808 000004df  a612                         LDA     #%00010010      WINDOW CYAN
809 000004e1 >b700                         STA     CCR1
810 000004e3 >b700                         STA     CCR2
811
812 000004e5 >1800                 SEC5    BSET    4,STAT
813 000004e7  a61e                         LDA     #30
814 000004e9 >040005                       BRSET   2,STAT4,S30     CHANNEL MODE ?
815 000004ec >000002                       BRSET   0,STAT6,S30     NO, 2-DIGIT PROG No. ENTRY ?
816 000004ef  a606                         LDA     #6              NO, SO 6 SECONDS ONLY
817 000004f1 >b700                 S30     STA     TMR
818 000004f3  81                           RTS
```

```
                                ********************************************************
                                *                                                      *
                                *       Bottom row analogue bar.                       *
                                *                                                      *
                                ********************************************************

000004f4  0e121113              CHAR    FCB     $0E,$12,$11,$13         BARGRAPH CHARACTERS

000004f8  636f6e74a2b2a9ac      ANCH    FCB     $63,$6F,$6E,$74,$A2,$B2,$A9,$AC          ANALOG
00000500  63b321f4f6efecf5              FCB     $63,$B3,$21,$F4,$F6,$EF,$EC,$F5          LOGOS

00000508 >b700                  ANOSD   STA     W3
0000050a >080041                        BRSET   4,STAT5,LOGO    ANALOGS
0000050d >1800                          BSET    4,STAT5         SET-UP SKIP FLAG
0000050f  5f                            CLRX
00000510  a67f                          LDA     #127
00000512 >cd0000                        JSR     OSDCLR          CLEAR ALL CHARACTERS
00000515  ae1d                          LDX     #29
00000517 >6fff                  COOP    CLR     CAS1-1,X
00000519  5a                            DECX
0000051a  26fb                          BNE     COOP

0000051c  a60a                          LDA     #%00001010      COLOR 2 = GREEN
0000051e  b733                          STA     C34             COLOR 3 = BLUE
00000520  a6e1                          LDA     #%11100001      OSD & PLL ON,
00000522  b735                          STA     WCR             WINDOW ON (COLUMN 1)
00000524  a622                          LDA     #%00100010      HORIZONTAL POSITION : TWO
00000526  b737                          STA     HPD
00000528  3f30                          CLR     $30             PUT A SPACE AT 17th AND 18th
0000052a  3f31                          CLR     $31             CHARACTERS
0000052c  a6e6                          LDA     #%11100110      COLOR 1,0 = RED, YELLOW, EDGE ON
0000052e >b700                          STA     CAS1            AND WINDOW ON (USING BIT 6)
00000530 >b700                          STA     CAS2
00000532  a6a6                          LDA     #%10100110      COLOR 1,0 = RED, YELLOW, WINDOW OFF
00000534 >b700                          STA     CAS3
00000536  a610                          LDA     #%00010000      SINGLE WIDTH/HIGHT, INTERRUPTS ON
00000538 >b700                          STA     RAD1
0000053a >b700                          STA     RAD2
0000053c >b700                          STA     RAD3
0000053e  a6e3                          LDA     #%11100011      WINDOW WHITE, OFF AT 3
00000540 >b700                          STA     CCR1
00000542 >b700                          STA     CCR2
00000544  a611                          LDA     #%00010001      WINDOW BLACK, OFF AT 17
00000546 >b700                          STA     CCR3

00000548 >3f00                          CLR     OSDL
0000054a  a605                          LDA     #LTAB2-LTAB0
0000054c >b700                          STA     LIND            SECOND TABLE
                                ********************************************************
                                *                                                      *
                                *       Analogue logos.                                *
                                *                                                      *
                                ********************************************************


0000054e >be00                  LOGO    LDX     ANAL
00000550 >d60000                        LDA     ANCH,X
00000553 >c70000                        STA     DRAM
00000556  5c                            INCX
00000557 >d60000                        LDA     ANCH,X
0000055a >c70001                        STA     DRAM+1
0000055d  5c                            INCX
0000055e >d60000                        LDA     ANCH,X
00000561 >c70010                        STA     DRAM+16
00000564  5c                            INCX
00000565 >d60000                        LDA     ANCH,X
00000568 >c70011                        STA     DRAM+17
```

```
888                                 ********************************************************
889                                 *                                                      *
890                                 *       Analogue bar.                                  *
891                                 *                                                      *
892                                 ********************************************************
893
894 0000056b >b600                          LDA     W3
895 0000056d >3f00                          CLR     W2
896 0000056f  44                            LSRA
897 00000570 >3900                          ROL     W2
898 00000572  44                            LSRA
899 00000573 >3900                          ROL     W2
900 00000575 >b700                          STA     W3
901 00000577  ae10                          LDX     #16
902 00000579  5a                    LRAN    DECX
903 0000057a >b300                          CPX     W3
904 0000057c  270a                          BEQ     STAR
905 0000057e  2204                          BHI     DOT
906 00000580  a614                          LDA     #$14            4
907 00000582  200d                          BRA     SKST
908 00000584  a60e                  DOT     LDA     #$0E            .
909 00000586  2009                          BRA     SKST
910 00000588 >bf00                  STAR    STX     W1
911 0000058a >be00                          LDX     W2
912 0000058c >d60000                        LDA     CHAR,X          ., 1, 2 OR 3
913 0000058f >be00                          LDX     W1
914 00000591 >d70020                SKST    STA     DRAM+32,X
915 00000594  5d                            TSTX
916 00000595  26e2                          BNE     LRAN
917
918 00000597 >cc0000                        JMP     SEC5
919
920                                         END
```